Technical Report No. 611

How to remove a dynamic prompt:

static and dynamic delimited continuation operators are equally expressible by Oleg Kiselyov^∗

March 2005

Computer Science Department Indiana University Bloomington, Indiana 47405-7104

∗Collaborating with Profs. Daniel P. Friedman and Amr A. Sabry on logical programming systems

Abstract

We show how to remove a dynamic prompt (aka reset) and thus to turn so-called static delimited continuation operators (shift/reset) into dynamic ones: control, control0, shift0. Our technique extends the continuation captured by shift by composing it with the previous fragments of the ‘stack’. Composition of context fragments can be done via regular functional composition. We thus demonstrate that all the above delimited continuation operators are macroexpressible in term of each other — without capturing undelimited continuations and without using mutable state. Furthermore, the operators shift, control, control0, shift0 are the members of a single parameterized family, and the standard CPS is sufficient to express their denotational semantics.

We give the simplest Scheme implementation of the dynamic control operators. We give a formal simulation proof that control realized through shift indeed has its standard reduction semantics.

1 Introduction

There have been proposed a number of delimited continuation operators, which are concisely surveyed in [8]. The most well-known operators are control/prompt (which were historically the first) and shift/reset. It is of considerable interest both to the implementors and the users to establish the relationship among various control operators. In one direction, it is obvious that shift/reset is macro-expressible [5] in terms of control/prompt. Whether control/prompt and similar operators (shift0, splitter, spawn, cupto, etc.) can be macro-expressed in terms of shift/reset has been for a long time an open question [1]. The so-called static delimited continuation operators shift/reset have the well-known denotational (CPS) semantics [4]. It was unknown if control/prompt and the other, so-called dynamic delimited continuation operators, have likewise CPS semantics — or they require extensions to the standard CPS, as sometimes claimed [7].

These open questions were answered by Chung-chieh Shan [8], who for the first time showed that the dynamic delimited continuation operators are macro-expressible in terms of shift/reset. This paper presents another, more uniform solution. We outline the general method of removing a dynamic prompt, and so reify the operation of composing frames of control stack. Contrary to the common belief [7], we do not need any special primitives for composing stack fragments other than the operation of capturing the fragments with shift/reset, which is expressible in standard CPS. Furthermore, we show that both static and dynamic delimited continuation operators are the members of a single family of delimited control, parameterized by simple functions specifying the composition of stack fragments. We demonstrate the simplest known Scheme implementation of control/prompt and other delimited control operators in terms of shift/reset. Finally, we formally prove that control implemented in terms of shift indeed satisfies its standard reduction semantics. Our proof method is general and can be used for proving other properties of delimited continuations.

The next section introduces notation and defines delimited continuation operators under investigation. Section 3 formulates the theorem of macroexpressivity of control/prompt in terms of shift/reset, and gives corollaries. One of them states that the delimited continuation operators are the members of the same family. We then give the Scheme implementation of the dynamic delimited continuation operators, and tests. In Section 5, we prove the macro-expressivity theorem. The proof demonstrates the composition of captured delimited continuations, which is achievable without any new primitives. We then conclude.

2 Notation

For simplicity, we will be considering delimited continuation operators in the context of untyped call-by-value lambda-calculus. Generalizations to richer call-by-value calculi are straightforward. Table 1 presents the syntax of the language. We write let x = e_1 in e_2 as an abbreviation for (λx.e_2)e_1. Table 2 gives the standard small-step operational semantics of the controlfree fragment of the language, by introducing the reduction relation e_1 ▷ e_2 and the evaluation context M[]. We write e_1 ▷^∗ e_2 for the composition of zero or more reduction steps. Intuitively, to reduce a term e, we find its representation in the form M[e_0] where e_0 is a redex and e_0 ▷ e_1. Then M[e_0] ▷ M[e_1]. As usual, e[v/x] means a capture-free substitution of value v for variable x in expression e.

A typical proposal [8] for treating delimited continuations as first class values defines syntactic forms to set the delimiter, and to capture and reify the continuation up to the delimiter. The delimiter setting-form is commonly e ::= Expression

v Value | (e_1e_2) Application | D[e] Set a delimiter | ηf.e Capture and reify

v ::= Value

x, y, f, g Variables | λx.e Abstraction

Figure 1: Syntax

C[] ::= (Delimited) Evaluation context [ ] Hole | (C[])e | v(C[]) Application

M[] ::= Arbitrary evaluation context

C[] | M[D[C[]]]

(λx.e)v ▷ e[v/x] (β-) Redex

Figure 2: Operational Semantics

called “prompt” or “reset”. We will write D[...] for a general prompt, and prompt ..., reset ... etc. for the delimiters of specific delimited continuation proposals. The capturing and reifying form is commonly called “shift” or “control”. Likewise, we will use the notation ηf.e for a generic form and shift f.e, control f.e etc. for a specific delimited continuation operator.

To account for delimited continuation operators, we augment the operational semantics in Table 2. The so-called “static” delimited continuation operators shift/reset are defined by the following reduction rules. M[reset v] ▷ M[v] (S1) M[reset C[shift f.e]] ▷ M[reset e^′] where e^′ = let f =λx.reset C[x] in e

(S2)

Here, C[] is the context that does not cross the delimiter’s (i.e., reset) boundary. In contrast, the context M[] may do so.

All these reduction rules do not handle the case when ηf.e occurs without a dynamically enclosing D[...]. We shall call a program that exhibits such a case to be “stuck on control.” We could have followed the original Danvy and Filinski’s proposal of assuming that the entire program is enclosed in a top-level D[...]. However, we chose to consider stuck on control expressions to be in error — as do several implementations of delimited continuations, e.g., Scheme48.

The most well-known of so-called “dynamic” delimited continuation operators (and historically the first delimited continuation operators proposed) are control/prompt:

M[prompt v] ▷ M[v] (C1) M[prompt C[control f.e]] ▷ M[prompt e^′] where e^′ = let f =λx.C[x] in e

(C2)

All various delimited continuation operators ηf.e capture the continuation up to the closest dynamically enclosing delimiter, and reify that continuation as a function. That function, bound to the variable f in the body e may be invoked in the body zero or more times. The reified delimited continuation may also be incorporated in the result of e, and invoked elsewhere in the program many times. The difference among the delimited continuation operators emerges when either the body e or the captured continuation themselves contain delimited continuation operators. The operator shift encloses the captured continuation in a delimiter, whereas control does not. Therefore, in the case of shift, the delimited continuation operators in C[] can capture only a part of that context, but no more. In contrast, in the case of control, delimited continuation operators embedded in C[] can capture continuations that span far beyond that context. Other such dynamic delimited continuation operators are possible, whose reduction rules eliminate even more delimiters:

M[reset_0 C[shift_0 f.e]] ▷ M[e^′] where e^′ = let f =λx.reset_0 C[x] in e

(S2’)

M[prompt_0 C[control_0 f.e]] ▷ M[e^′] where e^′ = let f =λx.C[x] in e

(C2’)

Both these operators do not preserve the original delimiter, permitting therefore the body of the delimited continuation operator to capture a part of M[]. The operator shift0 was briefly considered in [3], whereas control0 is equivalent to a single-prompt cupto [1].

3 Removing a dynamic prompt and expressing

control via shift

We introduce a sum (discriminated union) datatype H, similar to Haskell’s Either datatype:

H ::= H v_1v_2 | HV v Such a datatype is trivially realized in our calculus (cf., one of the Scheme implementations below): constructors H and HV become regular functions that yield abstractions. Slightly informally, we will call H v_1v_2 and HV v values themselves.

We introduce the following source-to-source transformation:

prompt e =⇒ h(reset HV e) control f.e =⇒ shift f^′.H (h^′ · f^′)(λf.e)

(PR)

where the functions h and h^′ are defined as follows:

h(H fx) = prompt xf h(HV x) = x (H)

h^′(H fx) = shift g.H (h^′ · g · f)x h^′(HV x) = x

(H’)

We also introduce the following transformation rule:

M[C[e]] =⇒ M[h^′(reset HV (C[e]))] ≡ M[B[C[e]]] (PR1) The notation B[e] is a mere shorthand for h^′(reset HV e). The context C[] in (PR1) may be empty.

It is straightforward to extend (PR) to expressions and contexts. We will write e^n⇀ e for the transformation of an expression e yielding e by applying (PR) and n-times the rule (PR1). When we apply the rule (PR1), the place to insert B[] is chosen non-deterministically. If the number of applications of (PR1) is irrelevant, we may write simply e ⇀ e. The properties of the transformation are summarized by the following Theorem.

Theorem 1. Prompt and control are macro-expressible in terms of shift and reset: applying the transformation (PR) to a program yields an observationally equivalent program. In particular, the transformations (PR) and (PR1) have the following properties: v is a value (T1)

That is, values are transformed to values. M[e] ⇀ M[e] (T2) Transformations (PR) and (PR1) are context independent. This is another statement of macro-expressibility.

if e_1 ▷ e_2 and e_1^n⇀ e_1 then ∃m. e_2^m⇀ e_2 and e_1 ▷^∗ e_2 (T3) That is, the transformed code simulates reductions. In particular:

if e ▷^∗ v then e^0⇀ e and e ▷^∗ v (T4) and if reductions of e diverge, so do reductions of e.

3.1 One family of delimited continuation operators

Corollary 1. Both static and dynamic delimited continuation operators are the members of a single family reset_{hr} e and shift_{hs} f.e parameterized by functions hr and hs:

reset e = reset_{hrstop} e shift f.e = shift_{hsstop} f.e prompt e = reset_{hrstop} e control f.e = shift_{hsprop} f.e

reset_0 e = reset_{hrprop} e shift_0 f.e = shift_{hsstop} f.e prompt_0 e = reset_{hrprop} e control_0 f.e = shift_{hsprop} f.e

where reset_{hr} e and shift_{hs} f.e and the functions hr_{stop}, hs_{stop}, hr_{prop}, and hs_{prop} are themselves macro-expressible in terms of shift and reset:

reset_{hr} e =⇒ hr(reset HV e) shift_{hs} f.e =⇒ shift f^′.H (hs · f^′)(λf.e)

hr_{stop}(H fx) = reset_{hrstop} xf hr_{stop}(HV x) = x hs_{stop} = hr_{stop}

hr_{prop}(H fx) = xf hr_{prop}(HV x) = x hs_{prop}(H fx) = shift g.H (hs_{prop} · g · f)x hs_{prop}(HV x) = x

3.2 Fischer-style CPS transform of dynamic delimited continuation operators

Corollary 2. Dynamic delimited continuation operators control/prompt and the others have a simple denotational semantics and a Fischer-style [6] CPS transform, induced by the CPS transform of shift/reset.

4 Scheme implementation of the dynamic delimited continuation operators

We shall treat the sum datatype H as an abstract datatype with two constructors H and HV and a deconstructor case-H. The datatype can be realized by the following pure lambda-terms in our calculus:

; Constructors
(define (H a b)
  (lambda (on-h)
    (lambda (on-hv)
      ((on-h a) b))))

(define (HV v)
  (lambda (on-h)
    (lambda (on-hv)
      (on-hv v))))

; Deconstructor
(define-syntax case-H
  (syntax-rules ()
    ((case-H e
       ((f x) on-h)
       (v on-hv))
     ((e (lambda (f) (lambda (x) on-h)))
       (lambda (v) on-hv)))))

Alternatively, we may, potentially more efficiently, implement the H datatype via Scheme’s indiscriminated union:

(define H-tag (list ’H-tag))

; Constructors
(define (H a b) (cons H-tag (cons a b)))
(define-syntax HV ; just the identity
  (syntax-rules ()
    ((HV v) v)))

; Deconstructor
(define-syntax case-H
  (syntax-rules ()
    ((case-H e
       ((f x) on-h)
       (v on-hv))
     (let ((val e))
       (if (and (pair? val) (eq? (car val) H-tag))
         (let ((f (cadr val)) (x (cddr val))) on-h)
         (let ((v val)) on-hv))))))

The family of the delimited continuation operators is implemented by the straightforward transcription into Scheme of the expressions in Corollary 1.

(define-syntax greset
  (syntax-rules ()
    ((greset hr e) (hr (reset (HV e))))))

(define-syntax gshift
  (syntax-rules ()
    ((gshift hs f e)
      (shift f* (H (lambda (x) (hs (f* x))) (lambda (f) e))))))

(define (hr-stop v)
  (case-H v
    ; on-h
    ((f x) (greset hr-stop (x f)))
    (v v))) ; on-hv

(define hs-stop hr-stop)

(define (hr-prop v)
  (case-H v
    ; on-h
    ((f x) (x f))
    ; on-hv
    (v v)))

(define (hs-prop v)
  (case-H v
    ; on-h
    ((f x)
      (shift g
        (H (lambda (y) (hs-prop (g (f y)))) x)))
    ; on-hv
    (v v)))

Using the operational semantics of shift/reset, it is easy to see that reset_{hrstop} e is reset e, and shift_{hsstop} f.e is itself shift f.e.

The operators prompt and control, in particular, have the following simple implementation:

(define-syntax prompt
  (syntax-rules ()
    ((prompt e) (greset hr-stop e))))

(define-syntax control
  (syntax-rules ()
    ((control f e) (gshift hs-prop f e))))

It is patently clear then that control and prompt are indeed macroexpressible in terms of shift and reset.

4.1 Tests

The tests run on Scheme48, with its built-in implementation of shift and

reset.

(reset (let ((x (shift f (cons ’a (f ’()))))) (shift g x)))
; ==> ’(a)
(prompt (let ((x (control f (cons ’a (f ’()))))) (control g x)))
; ==> ’()

(prompt ((lambda (x) (control l 2)) (control l (+ 1 (l 0)))))
; ==> 2
(prompt (control f (cons ’a (f ’()))))
; ==> ’(a)
(prompt (let ((x (control f (cons ’a (f ’())))))
          (control g (g x))))
; ==> ’(a)
(prompt0 (prompt0 (let ((x (control0 f (cons ’a (f ’())))))
                     (control0 g x))))
; ==> ’()
; prompt0 is the same as reset0
(prompt0 (cons ’a (prompt0 (shift0 f (shift0 g ’())))))
; ==> ’()
(prompt0 (cons ’a (prompt0 (prompt0 (shift0 f (shift0 g ’()))))))
; ==> ’(a)

; Example by Chung-chieh Shan
(prompt (begin (display (control f (begin (f 1) (f 2))))
               (display (control f (begin (f 3) (f 4))))))
; ==> 132342344234442344442344444234444442344444442344444444234....

5 Proof of Theorem 1

In this section we formally prove that control/prompt operators realized in terms of shift/reset, Theorem 1, satisfy their standard reduction semantics, eqs. (C1-C2). To be precise, we prove that reductions in the translated language (control/prompt expressed via shift/reset) simulate reductions in the original language. We must stress that our proof elucidates how the context fragment compositions (implicit in the original language with control/prompt) become explicit, as functional compositions, in the translated language.

Properties (T1) and (T2) are obvious. Before we prove the simulation property (T3), we consider the following Lemma.

Lemma 1.

B[v] ▷^∗ v

The proof is straightforward from the definition of B[e] as h^′(reset HV e), the fact that HV v is also a value, the definition of h^′ in eq. (H’), and eq. (S1). The Lemma is easily generalized to arbitrary sequences of B[].

The notation e_1 ▷ e_2 describes three different reductions: the βreduction in Table 2 and control reductions (C1) and (C2). We shall consider all three reductions in turn.

In order for the β-reduction to apply, the expression e_1 must have the form M[(λx.e^′_1)v]. Thus we have:

e_1 ≡ M[(λx.e^′_1)v] n⇀ M[(B[λx.e^′_1])(B[v])] { Lemma 1 } ▷^∗ M[(λx.e^′_1)v] ▷ M[e^′_1[v/x]]

On the other hand,

e_1 ≡ M[(λx.e^′_1)v] ▷ M[e^′_1[v/x]] m⇀ M[e^′_1[v/x]]

for some m.

Another possible reduction is (C1). It applies when the expression in question has the form M[prompt v]. Therefore,

e_1 ≡ M[prompt v] n⇀ M[h(reset HV (B[v]))] { Lemma 1 } ▷^∗ M[h(reset HV v)] { equations (S1) and (H) } ▷^∗ M[v]

The latter expression is obviously a transform image from M[v], which is the result of one-step reduction of e_1.

Lemma 2.

B[C[shift g.H (h^′ · g · f^′)(λf.e)]] ▷^∗ shift g.H (λx.h^′(g(B[C[f^′x]])))(λf.e) Informally, the translation image of control f.e can be lifted out of B[C[]] with the latter context fragment added to the captured continuation. Indeed,

B[C[shift g.H (h^′ · g · f^′)(λf.e)]] { Definition of B[] } ≡ h^′(reset HV (C[shift g.H (h^′ · g · f^′)(λf.e)])) { (S2), along with the fact HV e may be subsumed } { as a part of a non-reset-crossing context C[] } ▷ h^′(reset let g =λx.reset HV (C[x]) in H (h^′ · g · f^′)(λf.e)) { H ab is a value, apply (S1) and then (H’) } ▷^∗ let g =λx.reset HV (C[x]) in shift g^′.H (h^′ · g^′ · h^′ · g · f^′)(λf.e) ≡ shift g^′.H (λx.h^′(g^′(h^′(reset HV (C[f^′x])))))(λf.e) ≡ shift g^′.H (λx.h^′(g^′(B[C[f^′x]])))(λf.e)

The last reduction to consider, (C2), applies when the expression e_1 has the form of M[prompt C[control f.e]]. When we transform that expression, we have to keep in mind that the transform of a non–delimiter-crossing context C[] may, in general, be a delimiter-crossing context: a potentially inserted B[] contains the delimiter, reset. We first consider the case, however, when the image of the context C[] is a non–delimiter-crossing context C[]. We can then write:

e_1 ≡ M[prompt C[control f.e]] n⇀ M [ h(reset HV (C[shift f^′.H (h^′ · f^′)(λf.e)])) ] { (S2), along with the fact HV e may be subsumed } { as a part of a non-reset-crossing context C[] } ▷ M [ h(reset let f^′ =λx.reset HV (C[x]) in H (h^′ · f^′)(λf.e)) ] { H ab is a value, apply (S1) and then (H) } ▷^∗ M [ h(reset HV (let f^′ =λx.reset HV (C[x]) in (λf.e)(h^′ · f^′))) ] ≡ M [ h(reset HV (let f =λx.h^′(reset HV (C[x])) in e)) ] { definition of B[] } ≡ M [ h(reset HV (let f =λx.B [ C[x] ] in e)) ]

(*)

On the other hand,

e_1 ≡ M[prompt C[control f.e]] { apply (C2) } ▷ M[prompt let f =λx.C[x] in e] m⇀ M [ h(reset HV (let f =λx.B [ C[x] ] in e)) ] for some m — actually, for m = n + 1.

Finally, we consider the general case when the image of the context C[] does include B[]. We can always decompose C[] into C_2[C_1[]] so that C[] ≡ C_2[C_1[]]^n⇀ M_2 [ B [ C_1[] ]] (D) where either of the contexts C_1[] or C_2[] may be empty and C_2[]^m⇀ M_2[] for some m smaller than n. This equation merely expresses the fact of the insertion of B[]. It follows then:

e_1 ≡ M[prompt C[control f.e]] n⇀ M [ h(reset HV (M_2 [ B [ C_1[shift f^′.H (h^′ · f^′)(λf.e)] ]] )) ] { Lemma 2 } ▷^∗ M [ h(reset HV (M_2 [ shift f^′.H (λx.h^′(f^′(B [ C_1[x] ] )))(λf.e) ] )) ] (**)

If the context M_2[] does not cross the delimiter boundary — that is, contains no B[], we can denote it as C_2[] and apply (*), obtaining M [ h(reset HV (let f =λx.B [ C_2 [ B [ C_1[x] ]]] in e)) ] which is again the image of M[prompt let f =λx.C[x] in e] keeping in mind that C[e] ≡ C_2[C_1[e]] by (D). If, however, M_2[] did contain B[], we can apply (D) and (**) one more time. We recall that C_2[]^m⇀ M_2[] with m strictly less than n, thus our procedure must terminate.

Property (T4) is a simple consequence of (T3) and the fact v^n⇀ e ▷^∗ v, which follows from Lemma 1. We also observe that the transform of a stuckon-control expression is itself stuck on control.

6 Conclusions

We have introduced a general method of removing a dynamic prompt, so that delimited continuation operators may capture continuations beyond that prompt. Contrary to the previously held beliefs, composing of contexts can be done by the regular functional composition. No primitives other than the operation of capturing fragments of the context with shift/reset are needed. Therefore, the standard CPS is sufficient for giving denotational semantics of the dynamic delimited continuations, and the dynamic delimited continuation operators are just as expressible as shift/reset. We thus confirm the results of [8] using a more general framework and giving the formal proof.

Our technique of removing a dynamic prompt is rather general and thus can be used for the design of many other delimited control operators, which permit capturing the continuation up to an arbitrary (not necessarily the closest one) enclosing delimiter. It becomes entirely up to the user which delimiters to skip and how the captured frames are to be composed or used separately. The essence of our technique is reifying the control effect (or the absence of it) into regular values, instances of the datatype H. Therefore, the user program has the ability to determine which control effects, if any, had occurred during the evaluation of a piece of code — and act accordingly. Our technique is thus reminiscent of “referring to the central authority” approach by Cartwright and Felleisen [2]. However, instead of one central authority, we have bureaucracy — a sequence of authorities. A particular authority may either fulfill a request for service, or pass it, after perhaps modifying it, up the (dynamic) chain of hierarchy.

Practically, we have shown that both static and dynamic delimited continuation operators are the members of a single family of delimited control, parameterized by simple functions specifying the composition of stack fragments. This gives rise to the simplest Scheme realization of the delimited control operators in terms of shift/reset.

Acknowledgments

This paper owes its existence to [8]. The inspiration of the latter cannot be over-emphasized. I am very grateful to Chung-chieh Shan for many insightful discussions. I would like to thank Amr A. Sabry for his encouragement and shepherding, and Daniel P. Friedman for valuable comments.

References

[1] Carl A. Gunter, Didier Rémy, and Jon G. Riecke. 1995. A generalization of exceptions and control in ML-like languages. In Functional programming languages and computer architecture: 7th conference, ed. Simon L. Peyton Jones, 12–23. New York: ACM Press.

[2] Cartwright, Robert, and Matthias Felleisen. 1994. Extensible denotational language specifications. In TACS, ed. Masami Hagiya and John C. Mitchell, vol. 789 of Lecture Notes in Computer Science, 244–272. Springer.

[3] Danvy, Olivier, and Andrzej Filinski. 1989. A functional abstraction of typed contexts.

Tech. Rep. 89/12, DIKU, University of Copenhagen, Denmark.

[4] ———. 1990. Abstracting control. In Proceedings of the 1990 ACM conference on LISP and functional programming, 151–160. New York: ACM Press.

[5] Felleisen, Matthias. 1991. On the expressive power of programming languages. In Science of computer programming, vol. 17, 35–75. Preliminary version in: Proc. European Symposium on Programming, Lecture Notes in Computer Science, 432. Springer-Verlag (1990), 134–151.

[6] Fischer, Michael J. 1993. Lambda-calculus schemata. Lisp and Symbolic Computation 6(3-4):259–288.

[7] Matthias Felleisen, Daniel P. Friedman, Mitchell Wand, and Bruce F. Duba. 1988.

Abstract continuations: A mathematical semantics for handling full jumps. In Proceedings of the 1988 ACM conference on LISP and functional programming, 52–62. New York: ACM Press.

[8] Shan, Chung-chieh. 2004. Shift to control. In Proceedings of the 5th workshop on scheme and functional programming, ed. Olin Shivers and Oscar Waddell, 99–107. Technical report 600, Computer Science Department, Indiana University.